A GENERAL WINDING LAW OF MODULAR STELLARATOR COILS

E. Harmeyer, J. Kisslinger, F. Rau, H. Wobig

IPP 2/274

March 1985

MAX-PLANCK-INSTITUT FÜR PLASMAPHYSIK

8046 GARCHING BEI MÜNCHEN

MAX-PLANCK-INSTITUT FÜR PLASMAPHYSIK

GARCHING BEI MÜNCHEN

A GENERAL WINDING LAW OF MODULAR STELLARATOR COILS

E. Harmeyer, J. Kisslinger, F. Rau, H. Wobig

IPP 2/274 March 1985

Die nachstehende Arbeit wurde im Rahmen des Vertrages zwischen dem Max-Planck-Institut für Plasmaphysik und der Europäischen Atomgemeinschaft über die Zusammenarbeit auf dem Gebiete der Plasmaphysik durchgeführt.

A GENERAL WINDING LAW OF MODULAR STELLARATOR COILS

E.Harmeyer, J.Kisslinger, F.Rau, H.Wobig

ABSTRACT

A general winding law for twisted coils is described in terms of analytic functions. On a toroidal surface given as $\vec{x} = \vec{x}(\phi, \theta)$ (ϕ = toroidal coordinate, θ = poloidal coordinate) the current filaments $\phi = \phi_k(\theta)$ are represented in terms of sin - functions. The paper includes several examples of classical stellarator and HELIAC configurations. By specifying a separate winding law for every single coil also WVII - AS type configurations can be modelled. The numerical code describing the central filament and the borderlines of the coils are given in the appendix.

Introduction

The rotational transform in stellarators can be achieved either by helical windings or by modular twisted coils. Even with plane coils an appreciable rotational transform is obtained if the coils are arranged on a helical closed curve like in a FIGURE-8-stellarator or ASPERATOR. In classical stellarators with a circular magnetic axis plane coils only provide a small rotational transform[1] ,an azimutal elongation of the coils,however,as proposed in[2] increases the rotational transform appreciably. In general any stellarator field can be generated by poloidally closed current filaments. The magnetic potential Φ = const on a magnetic surface defines a surface current which produces the magnetic field inside the surface. Dommaschk[3] has develloped a method where the given field is represented in terms of Dommaschk functions. Solving a Neumann boundary value problem the magnetic potential Φ_i inside a toroidal surface S and Φ_a outside that surface are calculated. The desired surface current on S is defined by the isolines $\Phi_a - \Phi_i$ = const. By this method modular coils of the W VII-AS stellarator were constructed.

For parametric studies, however, it is more convenient to represent the modular coils in terms of an analytic winding law. A set of independent parameters $\alpha_1 .... \alpha_n$ characterizes the specific coil set. The number of parameters should be kept as small as possible.

Chapter I

In the following chapter a general winding law of modular stellarator coils is described which covers the classical $\ell = 2$ or $\ell = 3$ - stellarator as well as configurations with helical magnetic axes like HELIAC. The procedure starts with a helical and toroidally closed line which connects the centres of the single modular coils. In a cylindrical coordinate system (r, ϕ, z) this curve is written as follows

$$\vec{X}_o(\phi) = \vec{R}(\phi) + \vec{R}_H(\phi) \tag{1}$$

[1] S.N. Popov and A.P. Popryadukhin , Sov.Phys. - Tech.Phys. Vol 11 pp. 284 - 285 Aug.1966
[2] S. Rehker ,H. Wobig in Proc. 7th Symp. Fusion Tech. Grenoble pp. 345 - 357 Oct.1972
[3] W. Dommaschk Z. Naturforsch Vol 36a p.251 , 1981

with $\vec{R}(\phi)$ describing a circle ($R_o = const$)

$$\vec{R}(\phi) = \{ R_o \cos\phi, R_o \sin\phi, 0 \} \tag{2}$$

and

$$\vec{R}_H(\phi) = \{ a\cos\psi\cos\phi,\ a\cos\psi\sin\phi,\ b\sin\psi \} \tag{3}$$

describing ellipses around $\vec{R}(\phi)$. The poloidal angle ψ is correlated to the independent variable ϕ by

$$\psi = M\phi + S\sin M\phi \tag{4}$$

where M is the number of toroidal periods and S a number which controls the pitch of the central helix. The radius $r_H(\psi)$ of the helix is given by

$$r_H(\psi) = \sqrt{a^2\cos^2\psi + b^2\sin^2\psi} \tag{5}$$

a,b are constants.



FIG. 1: Coordinate system defining the central helix

In every point P of the central helix a cartesian coordinate system with the base vectors $\vec{t}(\phi)$,$\vec{b}(\phi)$,$\vec{n}(\phi)$ is attached which serves as a coordinate system for the modular coil. In special cases these base vectors may coincide with the tangent,binormal and normal vectors of the central helix but in general this coordinate system $\vec{t}, \vec{b}, \vec{n}$ is rotated with respect to the natural coordinate system. We choose $\vec{t}(\phi)$ to be the tangent vector of a second auxiliary helix which follows the same law (1)-(5) but with different parameters. The vector $\vec{b}$ is constructed in the following way : Let $\vec{a}(\phi)$ be a vector linearily independent of $\vec{t}(\phi)$,then $\vec{b}$ is defined by

$$\vec{b} = \frac{\vec{t} \times \vec{a}}{|\vec{t} \times \vec{a}|} \tag{6}$$

$\vec{n}$ is the vector product of $\vec{t}$ and $\vec{b}$. The $\vec{n}$ - $\vec{b}$ - plane is used to define the coil winding law.



FIG. 2: Intersection of the toroidal surface with the $\vec{n}$ - $\vec{b}$ - plane

The radius $r_o(\theta)$ of the coil is given in this $\vec{n}$ - $\vec{b}$ - plane as a function of the poloidal angle θ. $\theta = 0$ coincides with the $\vec{n}$-axis. In general $r_o(\theta, \phi)$ can also be a function of the toroidal angle ϕ. Thus a toroidal surface is obtained by

$$\vec{x}(\theta, \phi) = \vec{R}(\phi) + \vec{R}_H(\phi) + r_o \cos\theta\, \vec{n} + r_o \sin\theta\, \vec{b} \tag{7}$$

The curve $\vec{x}(\theta, \phi)$ with $\phi = const$ describes a plane closed curve which can be considered as the central filament of a coil. In case of circular coils we have $r_o = const$. In order to describe twisted modular coils a function

$$\phi = \phi_k(\theta) + \phi_k \tag{8}$$

has to be added. The toroidal surface given by (7) and the special winding law (8) for the k^{th} coil are the general frame to describe modular twisted coils. The radius of the coil is given in terms of a Fourier series

$$r_o(\theta, \phi) = r_{sp}\left(1 + F(A\cos\theta + B\cos 2\theta + C\cos 3\theta + D\cos 4\theta)\right) \quad (9)$$

The parameters r_{sp},F, A, B, C, D may be functions of the toroidal angle ϕ. The toroidal elongation of the twisted coil also can be represented in terms of a Fourier series.This method is inconvenient in case of a large twist because a large number of terms is needed. It is more appropriate to represent the curve $\phi_k(\theta)$ as a piecewise continuous curve with continuous derivatives.



FIG. 3: Angular winding law of twisted coils

$$\begin{aligned}
&\text{In region I} && \phi_1^k(\theta) = A_1 \sin\frac{\pi}{2}\frac{\theta}{\theta_o} \\
&\text{In region II} && \phi_2^k(\theta) = A_1 \sin\frac{\pi}{2(\pi-\theta_o)}(\theta + \pi - 2\theta_o) \qquad (10) \\
&\text{In region III} && \phi_3^k(\theta) = A_1 \sin\frac{\pi}{2}\frac{\theta - 2\pi}{\theta_o}
\end{aligned}$$

θ_o is a fixed angle describing the localisation of the effective toroidal current. The whole procedure as shown in the figure could be repeated several times in the intervall $[0, 2\pi]$ thus representing the classical $\ell = 1, 2$ or 3 - windings. Also several functions of this kind can be superimposed. The vector $\vec{a}$ needs further specifications. If $\vec{a}$ is choosen to be a non-rotating vector for example $\vec{e}_z$ or $\vec{R}$, the coil coordinate system $\vec{n}\,\vec{b}\,\vec{t}$ does not rotate around the $\vec{t}$ - direction. A rotating coordinate system is obtained, if $\vec{a}$ is identified with the helical vector $\vec{R}_H(\phi)$ of the central helix or the auxiliary helix.

CLASSICAL STELLARATOR (or TORSATRON/HELIOTRON).

In this case the central helix is a circle with constant radius R_o. The vector $\vec{t}$ is the unit vector in ϕ - direction and $\vec{a}$ is chosen to be the rotating vector $\vec{R}_H$ of the auxiliary helix. According to the specific choice of $r_o(\theta)$ and $\phi_k(\theta)$ $\ell = 2$ or 3 systems can be represented. M + S - effects are introduced by changing to a plane but noncircular central helix. The standard HELIAC requires a helical central line and $\vec{t}$ being the unit vector in ϕ - direction. The vector $\vec{a}$ is equal to $\vec{e}_z$ or $\vec{R}$. In this case the coil winding law is very simple :

$$r_o(\phi) = const, \; \phi = const$$

Also a toroidal circular conductor and vertical field coils are needed. Another simple case is the FIGURE -8 -STELLERATOR or ASPERATOR with circular coils. Here $\vec{t}$ is identical to the tangential vector of the central helix or the auxiliary helix. $\vec{a}$ is identified with $\vec{R}_H$. A small step further provides a modular HELIAC. By proper choice of $r_o(\theta)$ D - shaped or bean - shaped coils are modelled and the toroidal elongation is chosen according to eq.(10)

FINITE SIZE COILS

The curve $\vec{x} = \vec{x}(\theta, \phi_k(\theta)) =: \vec{x}_k(\theta)$ represents the central filament of the k-th twisted coil. For many magnetic field studies it may be sufficient to calculate the field from the central conductor only. If the field in the very neighbourhood of the coil is needed, however, the coil has to be represented by several filaments. For this purpose parallel curves to the central filament have to be constructed. The procedure is the following : Let $\vec{x}_k(\theta)$ be the central filament and $\vec{t}_k$ its tangential vector. If $\vec{n}_s$ is the normal vector of the toroidal surface $\vec{x}(\theta\,,\,\phi)$, we may use the base vectors $\vec{t}_k\;,\vec{n}_s\;,\vec{b}_k = \vec{t}_k \times \vec{n}_s$ in order to construct the borderline of the finite size coil. If h is the height of the coil and b the width the four borderlines are defined by

$$\vec{y}_k(\theta) \quad = \quad \begin{cases} \vec{x}_k(\theta) + \frac{b}{2}\vec{b}_k \pm \frac{h}{2}\vec{n}_s \\ \vec{x}_k(\theta) - \frac{b}{2}\vec{b}_k \pm \frac{h}{2}\vec{n}_s \end{cases} \qquad (11)$$

With other parameter than h and b several current filaments are defined which have a constant distance to the central current filament.

If the central current filament is a plane curve it is not appropriate to use the normal vector $\vec{n}_s$ as a base vector of the coordinate system. In this case it is better to use the tangential vector $\vec{t}_k$ and the normal vector $\vec{n}_k$ of the central filament as base vectors.



FIG. 4: Cross section and coordinate system of a coil

SUMMARY

The coil winding law given above allows to cover most of the stellarator configurations proposed so far. The parameters needed are

$$\begin{array}{rcl} \text{central helix} & : & R_0, a, b, M, S \\ \text{auxiliary helix} & : & R_0', a', b', M', S' \\ \text{toroidal surface} & : & r_{sp}, F, A, B, C, D \\ \text{coil winding law} \quad \phi_k & : & A_1,\ A_2,\ A_3,\ \theta_0, \ell \end{array} \tag{12}$$

In order to describe W VII - AS and related configurations, the parameters A_n ,F ,....D have to change from coil to coil. In chapter II several examples of modular stellerators are given. The figures exhibit a top view on the modular coil set and several pictures of the magnetic surfaces. The cross section of the magnetic surfaces is shown in a plane perpendicular to the magnetic axis. Also indicated is the cross section of the torus defining the coils (dashed line).

CHAPTER II

Several examples of standard stellarators of and two HELIAC configurations are given in the following chapter. The parameters used in the code are the following

M = number of field periods

NSP = number of coils per period

NSD = number of different data sets of coils

IQ = number of filaments per coil . Max IQ = 6. IQ =4 yields the four border lines

R0 = major radius

AC,BC,S = parameters of central helix.AC = a,BC=b

AP,BP,SP = parameters of auxiliary helix

AMR, SR = parameters of vector $\vec{a}$

$$\vec{a} = \quad \{\cos\psi\,\cos\phi,\, \cos\psi\,\sin\phi,\, \sin\psi\} \qquad \psi = \mathrm{AMR}\,\phi + \mathrm{SR}\,\sin \mathrm{AMR}\,\phi$$

3(A,N,T0) = parameters of angular winding law. A $=A_1, A_2, A_3$ = amplitude of ℓ = 1,2,3 - helical component. N = $\frac{24}{\pi}\theta_0$ defines the angle θ_0. T0 = initial phase angle of the toroidal elongation of the coils.($\phi_k = \phi_k(\theta - \mathrm{T0})$

PD , PZ = parameters describing tilting of coils.

$$\phi_k \quad = \quad \phi_k(\theta) + \mathrm{PZ}\,\sin(\theta)\,\cos(\mathrm{AMR}\,\phi) + \mathrm{PD}\,\sin(\theta)$$

PSH = toroidal angle of the first coil

R0S = r_{sp} = mean radius of coil

F,A,B,C,D, = parameters describing the radial coil winding law

D0,H0 = parameters defining the cross section of a coil. D0 = b/2 H0 = h/2. H1,D1 = parameters for wedge-shaped coils

STROM = current per coil in ampere turns.

AV1,AV2,AV3,AVF = perameters describing the toroidal variation of coil amplitudes

$$A_n(k^{th}\text{coil}) = A_n\left(1 + \text{AVn} \cdot \cos(\text{M} \cdot \phi_k)\right)$$

$$F_k = F(1 + \text{AVF} \cos(\text{M}\phi_k))$$

DELP1,DELP2 = parameters for toroidal angle ϕ_k of the coils

$$\phi_k = \text{PSH} + \delta\phi\left(k - 1 - \text{DELP1} \cdot \sin(\delta\phi(k-1) \cdot \text{M}) + \text{DELP2} \cdot \sin(\delta\phi(k-1) \cdot 2 \cdot \text{M})\right)$$

$$k = 1, 2, \ldots\text{NSP} \qquad \delta\phi = \frac{2\pi}{M \cdot \text{NSP}}$$

Six different examples of stellarators are listed on table 1. ℓ= 2- stellarator with 5 and 12 field periods, an $\ell = 3$ - stellarator with 6 field periods, a figure- 8 - stellarator with 8 field periods, a standard HELIAC with 4 field periods and a modular HELIAC with 5 field periods. The parameters of a WVII–AS -type configuration are listed separately.

Tab. 1.. LIST OF COIL PARAMETERS

	l=2 M=5	l=2 M=12	l=3 M=6	Asper. M=8	Heliac M=4	Hel.mod M=5
$R_0[cm]$	200	200	200	200	200	200
$a[cm]$	0	0	0	26	34	40
$b[cm]$	0	0	0	26	34	40
S	0	0	0	0	0	0
$a'[cm]$	0	0	0	26	0	40
$b'[cm]$	0	0	0	26	0	40
S'	0	0	0	0	0	0
SR	0	0	0	0	0	0
AMR	2.5	6	2	8	0	5
$r_{sp}[cm]$	40	44	40	24	50	40
F	0	1	0	0	0	0.53
A	–	0	–	–	–	0
B	–	-0.2	–	–	–	-0.45
C	–	0	–	–	–	0.15
D	–	0	–	–	–	-0.03
$A_1[rad]$	0	0	0	0	0	-0.075
$A_2[rad]$	0.06	0.07	0	0	0	0
$A_3[rad]$	0	0	-0.13	0	0	0
N_1	–	–	–	–	–	19
N_2	4	6	–	–	–	–
N_3	–	–	12	–	–	–
$PZ[rad]$	0.005	0	0	0	0	0.04
$PD[rad]$	0	0	0	0	0	0.12
$D0[cm]$	5	3	6	3	5	5
$H0[cm]$	8	8.4	9	3	8	8
AV_1	0	0	0	0	0	0
AV_2	0	0	0	0	0	0
AV_3	0	0	0	0	0	0
AVF	0	0	0	0	0	0
$DELP1$	0	0	0	0	0	0.24
$DELP2$	0	0	0	0	0	0



FIG. 5: $\ell = 2$ - STELLARATOR , M=5, $\iota(0) = 0.41$, $\iota(a) = 0.34$, aspect ratio A = 9, magnetic well 4%, one field period



FIG. 6: $\ell = 2$ STELLARATOR, M = 12 ,$\iota(0) = 0.28$, $\iota(a) = 0.7$, aspect ratio A = 7, magnetic well 3% in central region, two field periods



FIG. 7: $\ell = 3$ STELLARATOR, M = 6, $\iota = 0$, $\iota(a) = 0.7$, aspect ratio A = 7, marginal magnetic well



FIG. 8: HELIAC (with a central conductor and vertical field coils), M = 4 , $\iota(0) = 0.9$, $\iota(a) = 1.06$, A = 9 , magnetic well = 1%



FIG. 9: ASPERATOR-type configuration, M = 8, $\iota(0) = 2.1$ $\iota(a) = 2.2$, A = 11, magnetic hill 10 %, two field periods shown



FIG. 10: Modular HELIAC, M = 5, $\iota(0) = 1.97$
$\iota(a) = 1.94$,aspect ratio A = 10 , magnetic well 0.5 %



FIG. 11: A WVII - AS - type configuration, 5 field periods, $\iota(0) = 0.41$, $\iota(a) = 0.39$, aspect ratio $A = 11$, magnetic well 1 %

```
CMS    *   W7ASW CDATEN A1                         15/02/85    09:46:28     PAGE    1

  5   10   10    1                                                 M,NSP,NSD,IQ
  204.0     -6.       0.     -.00                                  R0,AC,BC,S
             -6.0      0.0     0.00    0.00   0.00                 AP,BP,SP,AMR,SR
 -.000 12   0.0 -.10   6 .36 -.002 12 .54  .0  -.00  .084 3(A,N,T0),PD,PZ,PHS
  44.0          6.0  0.0    9.0  0.0 -5.60E5                     R0S,D0,D1,H0,H1,STROM
  0.5    0.0    -.20    0.0    0.0    0.0                        F,A,B,C,D,E
 -.000 12   0.0 -.080 6 1.3 -.002 12 2.0  .0  -.00  .217 3(A,N,T0),PD,PZ,PHS
  46.0          6.0  0.0    9.0  0.0 -5.60E5                     R0S,D0,D1,H0,H1,STROM
  0.7    0.0    -.20    0.0    0.0    0.0                        F,A,B,C,D,E
 -.000 12   0.0 -.08   8 1.8 -.003 12 2.7  .0  -.00  .322 3(A,N,T0),PD,PZ,PHS
  48.0          6.0  0.0    9.0  0.0 -5.60E5                     R0S,D0,D1,H0,H1,STROM
  0.8    0.0    -.20    0.0    0.0    0.0                        F,A,B,C,D,E
 -.000 12   0.  -.070 10 2.4 -.003 12 3.6  .0  -.00  .428 3(A,N,T0),PD,PZ,PHS
  49.0          6.0  0.0    9.0  0.0 -5.60E5                     R0S,D0,D1,H0,H1,STROM
  0.9    0.0    -.20    0.0    0.0    0.0                        F,A,B,C,D,E
 0.000 12   0.0 .040 18 -.32 0.008 15 -.48  .0  .0  .573 3(A,N,T0),PD,PZ,PHS
  50.0          6.0  0.0    9.0  0.0 -5.60E5                     R0S,D0,D1,H0,H1,STROM
  1.0    0.0    -.20    0.0    0.0    0.0                        F,A,B,C,D,E
 -.000 12   0.0 .040 18  .32 0.008 15 -.48  .0  .0  .684 3(A,N,T0),PD,PZ,PHS
  50.0          6.0  0.0    9.0  0.0 -5.60E5                     R0S,D0,D1,H0,H1,STROM
  1.0    0.0    -.20    0.0    0.0    0.0                        F,A,B,C,D,E
 0.000 12  .0 -.070 10 -2.4 -.003 12 -3.6  .0  .00  .829 3(A,N,T0),PD,PZ,PHS
  49.0          6.0  0.0    9.0  0.0 -5.60E5                     R0S,D0,D1,H0,H1,STROM
  0.9    0.0    -.20    0.0    0.0    0.0                        F,A,B,C,D,E
 -.000 12   0.0 -.08 8 -1.8 -.003 12 -2.7  .0  .00  .935 3(A,N,T0),PD,PZ,PHS
  48.0          6.0  0.0    9.0  0.0 -5.60E5                     R0S,D0,D1,H0,H1,STROM
  0.8    0.0    -.20    0.0    0.0    0.0                        F,A,B,C,D,E
 -.000 12   0.  -.080 6 -1.3 -.002 12 -2.0  .0  .00 1.04 3(A,N,T0),PD,PZ,PHS
  46.0          6.0  0.0    9.0  0.0 -5.60E5                     R0S,D0,D1,H0,H1,STROM
  0.7    0.0    -.20    0.0    0.0    0.0                        F,A,B,C,D,E
 -.000 12   0.0 -.10 6 -.36 -.002 12 -.54  .0  .0 1.173 3(A,N,T0),PD,PZ,PHS
  44.0          6.0  0.0    9.0  0.0 -5.60E5                     R0S,D0,D1,H0,H1,STROM
  0.5    0.0    -.20    0.0    0.0    0.0                        F,A,B,C,D,E
```

```
CMS    *   ATFM CDATEN A1                          15/02/85    09:46:57     PAGE    1

 12    5    1    1                                                 M,NSP,NSD,IQ
  200.0     0.        0.     -.00                                  R0,AC,BC,S
            0.0      0.0     0.00    6.00   0.00                   AP,BP,SP,AMR,SR
 0.000 12   0.0  .070   6 0.0 -.0 12 .0 .00     -.00   .00 3(A,N,T0),PD,PZ,PSH
  44.0          3.0  0.0    8.4  0.0 -3.20E5                     R0S,D0,D1,H0,H1,STROM
 1.0  0.0  -.2  0.0  0.0  0.0                                      F,A,B,C,D,E
 -.00 0.0   0.  0.00 0.00 0.0                        AV1,AV2,AV3,AVF,DELP1,DELP2
```

```
C                                                    TEST.DECQHW        DE
      SUBROUTINE DECOUP                                                 DE
C                                                                       DE
C     DECOUP FUER 'TWISTED COILS'                                       DE
C     BILDUNGSGESETZ :                                                  DE
C     PHI = PHI0+A1*SIN(TS1)+A2*SIN(TS2)+A3*SIN(TS3)+PHD*COS(TETA) +    DE
C           +PHZ*SIN(TETA)                                              DE
C     A1,A2,A3 AMPLITUDENWERTE FUER L1,L2,L3 ANTEIL                     DE
C     TS1,TS2,TS3  WINKELWERTE    "  "  "  "  "                         DE
C     0<=N<=24,  N/L  GANZZAHLIG                                        DE
C     RK   MITTLERER SPULENRADIUS                                       DE
C     PHD   DREHWINKEL UM Z-ACHSE BEI R0(GROSSER RADIUS)                DE
C     NSP  ANZAHL DER SPULEN PRO FELDPERIODE                            DE
C     NSD  ANZAHL DER EINZULESENDEN SPULENDATENSAETZE                   DE
C     NSD .EQ. 1 NUR DATEN EINER SPULE WERDEN EINGELESEN,DIE DATEN DER  DE
C     ANDEREN SPULEN WERDEN BERECHNET                                   DE
C     IQ   ANZAHL DER STROMFAEDEN PRO SPULE                             DE
C     DELP1,DELP2   FAKTOREN FUER SPULENABSTANSVARIATION IN PHI         DE
C     PHZ      FAKTOR FUER VARIATION DER SPULENNEIGUNG UM BZ ZU ERZEUGEN DE
C     D1 FAKTOR FUER TRAPEZFOERMIGKEIT D. SPULEN BI=D0(1+D1),BA=D0(1-D1)DE
C     H1  OFFSET FUER DIE LAGE DER SPULENKANTEN ZUM ZENTRALEN FADEN     DE
C         HI = H0-H1,HA = H0+H1                                         DE
C     D0  HALBE DICKE DER SPULE                                         DE
C     H0  HALBE HOHE  DER SPULE                                         DE
C                                                                       DE
      LOGICAL RUN,RUNH,RUNF                                             DE
C                                                                       DE
      COMMON /BOUCLE/ NH1,NH2                                           DE
      COMMON /HH/     XP(51,300), YP(51,300), ZP(51,300), COURH(300),   DE
     1               A1, A2,   A3 ,FC, ASE,BSE,SSE,AMC,R0C,R0SC,        DE
     2                NSDC,NSP,MC,NH(300)                               DE
      COMMON /KK/     NHELI1,NHELI2,NHSOM,NPOLY1,NPOLY2,NPOSOM,NPAQZ1,  DE
     1                NPAQZ2,NSZSOM,NSPIR1,NSPIR2,NSPSOM,DMIN2,ERMAX2   DE
      COMMON /CSPUDI/ HI(6),DI(6),COURI(6),D0,D1,H0,H1,COUR,IQ          DE
      COMMON /CSEELE/ A,B,S,AMR,R0,DU1,PHI,DU2,M,RUN                    DE
      COMMON /CFDF/  Q,R0S,F,OX,OY,OZ,PX,PY,PZ,QX,QY,QZ,IA,RUNF         DE
      COMMON /FDC2/ RRA,RZA,R1S,ELR,MA,RUNH,RHV1                        DE
      COMMON        EKX(49),EKY(49),EKZ(49),BKX(49),BKY(49),BKZ(49)     DE
     &              ,N(3),T0(3),TS(48,3),SINTE(48)                      DE
      COMMON /CKREIS/ X(3),Y(3),Z(3),XMP,YMP,ZMP,XN,YN,ZN,XL,YL,ZL,RK   DE
      COMMON /CSPUL/   F1,AVF,NSD                                       DE
C                                                                       DE
      PARAMETER (PI=3.1415926536,PI2=2*PI,PIH=0.5*PI,DTETA=0.1309)      DE
C                                                                       DE
      RUNH =.FALSE.                                                     DE
      READ (5,*) M,NSP,NSD,IQ                                           DE
      IF (IQ.LE.0 .OR. IQ.GT.6) THEN                                    DE
      IQ = 1                                                            DE
      WRITE (6,6011)                                                    DE
      ENDIF                                                             DE
      NSQ = NSP*IQ                                                      DE
      NHSOM = NSQ*M                                                     DE
      IF(NHSOM.GT.300) THEN                                             DE
      WRITE(6,1050) NHSOM                                               DE
      STOP                                                              DE
```

```
      ENDIF
      RUN = .FALSE.
      PHI = 0.0
      Q = 0.0
      CALL SEELEH
      CALL PLAN
      WRITE (6,1020) M,NSP,IQ,NHSOM
      WRITE (6,1030)
      DPHI = PI2/M
      DPHIH= DPHI/NSP
      AM = AMR
      IF (AMR/ FLOAT(M).LT.0.49) AM = 0.0
C
      IF(NSD.EQ.1) THEN
      PHIS1 = 0.0
      READ (5,*) A11,N(1),T0(1),A21,N(2),T0(2),A31,N(3),T0(3),PHD,PHZ1
     &           ,PHIS2
      READ (5,*) ROS,D0,D1,H0,H1,COUR
      RUNF = .FALSE.
      CALL FDEFH
      F1 = F
      READ (5,*) AV1,AV2,AV3,AVF,DELP1,DELP2
      ENDIF
C
      DO 40 IP=1,48
   40 SINTE(IP) = SIN((IP-1)*DTETA)
C
      DO  50 INP=1,NSP
        IH = INP*IQ
        IF(NSD.EQ.1) THEN
        PHIS = PHIS1+DPHIH*(INP-1-DELP1*SIN(DPHIH*(INP-1)*M)
     &       + DELP1*DELP2*DSIN(DPHIH*(INP-1)*M*2))
        COSMP = COS(M*PHIS)
        SINMP = SIN(M*PHIS*0.5)
        A1 = A11*(1.0 + AV1*COSMP)
        A2 = A21*(1.0 + AV2*COSMP)
        A3 = A31*(1.0 + AV3*COSMP)
        F  = F1*(1.0 + AVF*COSMP)
        PHZ = PHZ1*COS(AM*PHIS) + PHIS2*SIN(M*0.5*PHIS)
      ELSE
      READ (5,*) A1,N(1),T0(1),A2,N(2),T0(2),A3,N(3),T0(3),PHD,PHZ,PHIS
      READ (5,*) ROS,D0,D1,H0,H1,COUR
      RUNF = .FALSE.
      CALL FDEFH
      ENDIF
      IA = 1
      IF(A1.EQ.0.0 .AND. A2.EQ.0.0 .AND. A3.EQ.0.0) IA = 0
C
      WRITE(6,1040) A1,N(1),T0(1),A2,N(2),T0(2),A3,N(3),T0(3),PHD,PHZ
     &           ,PHIS,ROS,D0,D1,H0,H1,COUR
C
      DO 100 L=1,3
        IF(N(L).LE.0 .OR. N(L).GE.24) THEN
       WRITE(6,1070) L,INP,N(L)
        STOP
```

```
      ENDIF
       DT1 = PIH/N(L)
       DT2 = PIH/(24 - N(L))
C
       N1H = 24/L
       N1 = 2*N1H
       NT0 = MOD(IFIX(T0(L)*N1/PI2 + SIGN(0.5,T0(L))),N1)
       I = 0
      DO 110 IP=1,24,L
       I = I+1
       I1 = MOD(I-1+NT0+N1,N1) + 1
       I2 = MOD(I1-1+N1H,N1) + 1
       TS(I1,L) = SIN(MIN0(N(L),IP-1)*DT1 + MAX0(IP-N(L)-1,0)*DT2)
  110  TS(I2,L)= SIN(PI+MIN0(24-N(L),IP-1)*DT2 + MAX0(IP+N(L)-25,0)*DT1)
  100  CONTINUE
C
      DO 102 IP=1,48
       IP2 = MOD(IP-1,24) + 1
       IP3 = MOD(IP-1,16) + 1
       Q   = DTETA*(IP-1)
       PHI = PHIS + A1*TS(IP,1) + A2*TS(IP2,2) + A3*TS(IP3,3)
     &      + (PHD+PHZ)*SINTE(IP)
       CALL SEELEH
       CALL PLAN
       CALL FDEFH
       XP(IP,IH) = OX
       YP(IP,IH) = OY
       ZP(IP,IH) = OZ
       EKX(IP) = PY*QZ - QY*PZ
       EKY(IP) = PZ*QX - QZ*PX
  102  EKZ(IP) = PX*QY - QX*PY
       XP(49,IH) = XP(1,IH)
       YP(49,IH) = YP(1,IH)
       ZP(49,IH) = ZP(1,IH)
C
       CALL SPUDI
C
       COURH(IH) = COUR
       NH(IH) = 48
       IF(IQ.NE.1) THEN
       DO 103 IP=1,48
       IPM1 = MOD(IP+46,48) + 1
       TX = XP(IP+1,IH) - XP(IPM1,IH)
       TY = YP(IP+1,IH) - YP(IPM1,IH)
       TZ = ZP(IP+1,IH) - ZP(IPM1,IH)
       BKX(IP) =-TY*EKZ(IP) + TZ*EKY(IP)
       BKY(IP) =-TZ*EKX(IP) + TX*EKZ(IP)
       BKZ(IP) =-TX*EKY(IP) + TY*EKX(IP)
       VR = 1.0/SQRT(BKX(IP)*BKX(IP) + BKY(IP)*BKY(IP) +BKZ(IP)*BKZ(IP))
       BKX(IP) = BKX(IP)*VR
       BKY(IP) = BKY(IP)*VR
       BKZ(IP) = BKZ(IP)*VR
       VR = 1.0/SQRT(EKX(IP)*EKX(IP) + EKY(IP)*EKY(IP) +EKZ(IP)*EKZ(IP))
       EKX(IP) = EKX(IP)*VR
       EKY(IP) = EKY(IP)*VR
```

```
      EKZ(IP) = EKZ(IP)*VR
  103 CONTINUE
      DO 105 LL=1,IQ
      IH1 = IH - IQ + LL
      COURH(IH1) = COURI(LL)
      NH(IH1) = 48
      DO 104 IP=1,48
      XP(IP,IH1) = XP(IP,IH) + DI(LL)*BKX(IP) + HI(LL)*EKX(IP)
      YP(IP,IH1) = YP(IP,IH) + DI(LL)*BKY(IP) + HI(LL)*EKY(IP)
      ZP(IP,IH1) = ZP(IP,IH) + DI(LL)*BKZ(IP) + HI(LL)*EKZ(IP)
  104 CONTINUE
      XP(49,IH1) = XP(1,IH1)
      YP(49,IH1) = YP(1,IH1)
      ZP(49,IH1) = ZP(1,IH1)
  105 CONTINUE
      ENDIF
   50 CONTINUE
C
      MC = M
      NSDC = NSD
      ASE = A
      BSE = B
      SSE = S
      AMC = AMR
      ROC = RO
      ROSC = ROS
      FC = F
C
      IF(M.NE.1) THEN
      DO 106 J =2,M
       SINP = SIN((J-1)*DPHI)
       COSP = COS((J-1)*DPHI)
      NSQ = NSP*IQ
      DO 106 INP=1,NSQ
       IHM = (J-1)*NSQ+INP
       COURH(IHM) = COURH(INP)
       NH(IHM) = NH(INP)
       DO 106 IP=1,49
       XP(IP,IHM) = XP(IP,INP)*COSP - YP(IP,INP)*SINP
       YP(IP,IHM) = YP(IP,INP)*COSP + XP(IP,INP)*SINP
       ZP(IP,IHM) = ZP(IP,INP)
  106 CONTINUE
      ENDIF
 1020 FORMAT(/,5X,'PERIODENZAHL',I4,10X,'SPULEN PRO PERIODE',I4,10X,
     1    'STROMFAEDEN PRO SPULE ',I4,
     2                           5X,'GESAMTZAHL DER STROMFAEDEN',I8)
 1030 FORMAT(/,5X,'A1',3X,'N1',5X,'T01',4X,'A2',3X,'N2',5X,'T02',4X,'A3'
     &,3X,'N3',5X,'T03',5X,'PHD',5X,'PHZ',5X,'PHIS',3X,'ROS',2X'H.BREITE
     & E',' D1',3X,'H.HOEHE    H1          COUR')
 1040 FORMAT(1X,F8.4,I3,2F8.4,I3,2F8.4,I3,4F8.4,F7.2,F8.2,F7.2,F8.2,F7.2
     &        ,E11.3)
 1050 FORMAT(1X,'FEHLER IN DER ANZAHL DER SPULEN MAX.300 ANZAHL=',I4)
 1070 FORMAT(1X,'N(',2I4,') NICHT IM RICHTIGEN BEREICH N=',I5)
 6011 FORMAT(1X,'IQ<1 ODER IQ>6,IQ WIRD AUF 1 GESETZT')
      RETURN
      END
```

```
C                                                    JOK:TEST.FDEFH        FD
      SUBROUTINE FDEFH                                                     FD
      LOGICAL RUN,RUNH,RUN1                                                FD
      COMMON /CFDF/  Q,ROS,F,OX,OY,OZ,PX,PY,PZ,QX,QY,QZ,IA,RUN             FD
      COMMON /CDERP/ DXN,DYN,DZN,DXL,DYL,DZL,TX,TY,TZ                      FD
      COMMON /CKREIS/ X(3),Y(3),Z(3),XMP,YMP,ZMP,XN,YN,ZN,XL,YL,ZL,RK      FD
      COMMON /CSEELE/ AS,BS,S,SD,R0,P(3),M,RUNH                            FD
      DATA RUN1 /.TRUE./                                                   FD
      SAVE A,B,C,D,E,RUN1                                                  FD
C                                                                          FD
C     FDEFH BERECHNET ORTSKOORDINATEN (OX,OY,OZ) UND ABLEITUNGEN EINER     FD
C     FLAECHE GEGEBEN DURCH RADIUS RS=F(Q) UND EINEM ORTSPUNKT (X,Y,Z)     FD
C     DER IN SUBR. SEELEH BERECHNET WIRD.                                  FD
C                                                                          FD
C      PARAMETER:                                                          FD
C    A,B,C,D,E KOEFFIZIENTEN ZUR DARSTELLUNG VON RS=F(Q)                   FD
C     F   MULTIPLIKATIONSFAKTOR FUER A,B,C,D,E                             FD
C     ROS MITTLERER SPULENRADIUS                                           FD
C                                                                          FD
      IF(RUN ) GOTO 11                                                     FD
      READ(5,*) F,A,B,C,D,E                                                FD
      IF (RUN1) WRITE(6,1000)                                              FD
      RUN1 = .FALSE.                                                       FD
       WRITE(6,1010) F,A,B,C,D,E                                           FD
      RUN = .TRUE.                                                         FD
C                                                                          FD
   11 SINQ =  SIN(Q)                                                       FD
      COSQ =  COS(Q)                                                       FD
      RS = ROS*(1+F*(A* COSQ  +B* COS(2.*Q)+C* COS(3.*Q)+D* COS(4.*Q)+     FD
     &                E* COS(5*Q)))                                        FD
      RSQ= ROS*F*(-A* SINQ  -2.*B* SIN(2.*Q)-3.*C* SIN(3.*Q)-4.*D*         FD
     &        SIN(4.*Q)-5.*E* SIN(5.*Q))                                   FD
      VL = RS*COSQ/RK                                                      FD
      BETM =  SQRT(XN*XN + YN*YN + ZN*ZN)                                  FD
      VM = RS*SINQ/BETM                                                    FD
      OX = X(2) + VL*XL + VM*XN                                            FD
      OY = Y(2) + VL*YL + VM*YN                                            FD
      OZ = Z(2) + VL*ZL + VM*ZN                                            FD
      VAR1 = (RSQ*COSQ - RS*SINQ)/RK                                       FD
      VAR2 = (RSQ*SINQ + RS*COSQ)/BETM                                     FD
      QX = XL*VAR1 + XN*VAR2                                               FD
      QY = YL*VAR1 + YN*VAR2                                               FD
      QZ = ZL*VAR1 + ZN*VAR2                                               FD
      IF(IA.NE.0) THEN                                                     FD
      DP1 = 1.0/(P(3)-P(1))                                                FD
      VAR1 = SQRT(DXL*DXL + DYL*DYL +DZL*DZL)                              FD
      VAR2 = SQRT(DXN*DXN + DYN*DYN +DZN*DZN)                              FD
      IF(VAR1.NE.0.0) VAR1 = RS*COSQ/VAR1                                  FD
      IF(VAR2.NE.0.0) VAR2 = RS*SINQ/VAR2                                  FD
      PX = (X(3)-X(1))*DP1 + VAR1*DXL + VAR2*DXN                           FD
      PY = (Y(3)-Y(1))*DP1 + VAR1*DYL + VAR2*DYN                           FD
      PZ = (Z(3)-Z(1))*DP1 +  VAR1*DZL + VAR2*DZN                          FD
      ELSE                                                                 FD
      PX = TX                                                              FD
      PY = TY                                                              FD
```

```
         PZ = TZ                                                          FD
         ENDIF                                                            FD
C                                                                         FD
 1000 FORMAT (/2X,'FDEFH  JOK:TEST.FDEFH  VOM 16.5.84'                    FD
     &,/,2X,'RS = R0S*(1+F*(A*COS(Q)+B*COS(2*Q)+C*COS(3*Q)+D*COS(4*Q)'    FD
     &,'+E*COS(5*Q)))')                                                   FD
 1010 FORMAT(5X,'F =',F8.4,5X,'A =',F8.4,5X,'B =',F8.4,5X,'C =',F8.4,5X,FD
     &   'D = ',F8.4,5X,'E =',F8.4                         )              FD
C                                                                         FD
         RETURN                                                           FD
         END                                                              FD


C                                                   JOK:TEST.SEELEH
      SUBROUTINE SEELEH
      LOGICAL RUN
      COMMON /CSEELE/ A,B,S,AMR,R0,P(3),M,RUN
      COMMON /CKREIS/ X(3),Y(3),Z(3),XMP,YMP,ZMP,XN,YN,ZN,XL,YL,ZL,RK
      DATA DP,EPS /0.002,1.0/
C
C     SEELEH BERECHNET AUS DEM G.RADIUS R0, DEM HELIKALEN RADIUS
C     RH = F(A,B,PSI) UND DER STEIGUNG S DIE KOORDINATEN DER
C     SPULENZENTREN  AM  ORT PHI, SOWIE AN DEN ORTEN PHI+-DP.
C
      IF(RUN ) GOTO 11
      READ(5,*) R0,A,B,S
       WRITE(6,1000) R0,A,B,S
      RUN = .TRUE.
C
   11 P(1) = P(2)- DP
      P(3) = P(2)+ DP
C
      DO 1 I=1,3
       PSI = M* P(I) + S*SIN(M*P(I))
       COSPSI = COS(PSI)
       SINPSI = SIN(PSI)
       X(I) = (R0 +  A*COSPSI)*COS(P(I))
       Y(I) = (R0 +  A*COSPSI)*SIN(P(I))
    1  Z(I) =   B*SINPSI
C
 1000 FORMAT (/2X,'SEELEH JOK:TEST.SEELEH VOM 5.2.85'
     &,/,2X,'X = (R0+A*COS(PSI))COS(PHI) , Y = (R0+B*COS(PSI))SIN(PHI),'
     &,'Z = B*SIN(PSI)',/,1X,'PSI = M*PHI + S*SIN(M*PHI)',/
     &          ,1X,'R0=',F8.4,5X,'A =',F8.4,5X,'B =',F8.4,5X,'S =',F8.4)
C
       RETURN
       END
```

```
C                                                      JOK:TEST.PLAN
      SUBROUTINE PLAN
C
C     ES WIRD DER NORMALENVECTOR N(XN,YN,ZN) UND DER BINORMALENVETOR
C     L(XL,YL,ZL) BERECHNET. N UND L STEHEN SENKRECHT AUF DER TANGENTE
C     DER SPULENSEELE. L GIBT DIE RICHTUNG FUER TETA=0 AN.
C
      LOGICAL RUNP
      COMMON /CDERP/ DXN,DYN,DZN,DXL,DYL,DZL,TX,TY,TZ
      COMMON /CPLAN/ AP,BP,SP,SR,AMR,RUNP
      COMMON /CSEELE/ A,B,S,AMC,R0,P(3),M,RUN
      COMMON /CKREIS/XX(3),YY(3),ZZ(3),XMP,YMP,ZMP,XN,YN,ZN,XL,YL,ZL,RK
      DATA RUNP/.FALSE./
C
      IF(RUNP) GOTO 11
      READ(5,*) AP,BP,SP,AMR,SR
       WRITE(6,1000) AP,BP,SP,AMR,SR
      RUNP= .TRUE.
      AMC = AMR
C
   11  SDSMP= SD*SIN(M*P(2))
       PSI = M* P(2) + SDSMP
       COSPSI = COS(PSI)
       SINPSI = SIN(PSI)
       SINP = SIN(P(2))
       COSP = COS(P(2))
       DPSIDP = M + M*SD*COS(M*P(2))
       DRDP = -AP*SINPSI*DPSIDP
       XR = (R0 + AP*COSPSI)*COSP
       YR = (R0 + AP*COSPSI)*SINP
       ZR = BP*SINPSI
C
      TX = -YR + COSP*DRDP
      TY =  XR + SINP*DRDP
      TZ = BP*COSPSI*DPSIDP
      DPSIPP = -M*M*SDSMP
      DRDPP = -AP*(COSPSI*DPSIDP*DPSIDP+ SINPSI*DPSIPP)
      DTX = -TY - SINP*DRDP + COSP*DRDPP
      DTY =  TX + COSP*DRDP + SINP*DRDPP
      DTZ =  BP*( -SINPSI*DPSIDP*DPSIDP + COSPSI*DPSIPP)
C
      PSI = AMR*P(2) + SR*SIN(AMR*P(2))
      COSPSI = COS(PSI)
      SINPSI = SIN(PSI)
      DPSIDP = AMR + AMR*SR*COS(AMR*P(2))
      XR = COSPSI*COSP
      YR = COSPSI*SINP
      ZR = SINPSI
      DXR = -YR - COSP*SINPSI*DPSIDP
      DYR = XR - SINP*SINPSI*DPSIDP
      DZR = COSPSI*DPSIDP
C
      XN = YR*TZ - ZR*TY
      YN = ZR*TX - XR*TZ
      ZN = XR*TY - YR*TX
```

```
      XL  = TY*ZN      - TZ*YN
      YL  = TZ*XN      - TX*ZN
      ZL  = TX*YN      - TY*XN
      RK  = SQRT(XL*XL + YL*YL + ZL*ZL)
      DXN = DYR*TZ + DTZ*YR - DZR*TY - ZR*DTY
      DYN = DZR*TX + DTX*ZR - DXR*TZ - XR*DTZ
      DZN = DXR*TY + DTY*XR - DYR*TX - YR*DTX
      IF (AMR.EQ.0.0) DZN = 0.0
      DXL = DTY*ZN + TY*DZN - DTZ*YN - TZ*DYN
      DYL = DTZ*XN + TZ*DXN - DTX*ZN - TX*DZN
      DZL = DTX*YN + TX*DYN - DTY*XN - TY*DXN
      RETURN
 1000 FORMAT(1X,'PLAN VOM 5.2.85   ',
     &    /,'AP =',F8.4,5X,'BP =',F8.4,5X,'SP =',F8.4,5X,'AMR =',F8.4,
     &   5X,'SR = ',F8.4)
      END
```